


Ideal Transformer

Transformer



Description:

This element implements an ideal transformer model, for a given turns ratio, n1 (primary number of turns) and n2 (secondary number of turns), where R is leakage resistance.

Form: transformer: <instance name> n₁ n₂ n₃ n₄ <parameter list>

n₁ port #1 signal input terminal

n₂ port #1 reference terminal

n₃ port #2 signal output terminal

n₄ port #2 reference terminal

Parameters:

Parameter	Type	Default value	Required?
n1: number of turns in primary	DOUBLE	1	No
n2: number of turns in secondary	DOUBLE	1	No
R: Leakage resistance	DOUBLE	1e10	No

Examples:

```
//          Vin  GND  Vout  GND
transformer:trf    2    0    3    0
* Where '2 0' input port & '3 0' output port with ground as references.
```

// Example for Step-Down Voltage

```
transformer:trf    2    0    3    0    n1=10
```

// Example for Step-Up Voltage

```
transformer:trf    2    0    3    0    n2=10
```

Model Documentation:

An ideal transformer is represented by independent sides of primary and secondary wings. The mathematical model of the element consists of voltage ratio that depends on turns in primary to

the secondary. The shunt leakage resistances were added to improve condition numbers for the simulator.

Let turn ratio be $t = \frac{n_2}{n_1}$ and $g=1/R$, then the following voltage relationship describes 2 sides of a transformer:

$$g(V_3 - V_4) = gt(V_1 - V_2) \quad \rightarrow \quad 0 = gtV_1 - gtV_2 - gV_3 + gV_4$$

With introduction of the other state variable, I, relationship of currents as follow:

$$\begin{aligned} I_1 &= gI \\ I_2 &= -gI \\ tI_3 &= -g * I \rightarrow I_3 = (-g / t) * I \\ tI_4 &= g * I \rightarrow I_4 = (g / t) * I \end{aligned}$$

The ideal transformer does not have admittance matrix, thus modified nodal sparse technique was used to create matrix. The leakage resistance was entered into admittance matrix. Based on voltage and currents relationship the following stamp was developed.

$$[M] = \left[\begin{array}{cc|cc|c} g & -g & & & g \\ -g & g & & & -g \\ & & g & -g & -g/t \\ & & -g & g & g/t \\ \hline tg & -tg & -g & g & 0 \end{array} \right] x \begin{bmatrix} V_1 \\ V_2 \\ V_3 \\ V_4 \\ I \end{bmatrix} = \begin{bmatrix} I_1 \\ I_2 \\ I_3 \\ I_4 \\ 0 \end{bmatrix}$$

This matrix was used to code the stamp for an ideal transformer.

References:

“Illustrated Guide to PSPICE”, Robert Lamey, ISBN: 0827365241

“The Electrical Engineering Handbook”, Wai-Kai Chen, ISBN: 0121709604; pp55-56

“Fundamental Electrical Technology”, Marvin Klayton, ISBN: 0201038307

Professor: Michael Steer

Sample Netlist: Step up example

```
***** Ideal Transformer Test *****
***** transformer.net
***** transformer In LRT Out LRT n1=turns_primary n2=turns_second

.options gmin=0 ftol=1.e-10

.tran2 tstop = 50e-3 tstep = 10e-6 gcomp=0 im=2
```

```

***** print detailed info to .OUT file
.options verbose

***** Circuit *****
*++++++ source for transient analysis
vsource:v1 1 0 vac=10V f=60
r:rin 1 2 r=50

*++++++for turn ratio 1-to-10 -> Vin/t=Vout: 10V=100V
transformer:t1 2 0 3 0 n2=10

r:rload 3 0 r=1e6

***** Simulation *****
.options gnuplot

***** Transient Simulation Results *****
*++++++ set up plot preamble with font and label information
*.options plotVT1Preamble="set term x11 font 'helvetica,18';
*       set title 'Source Voltage';
*       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage *(V_ac)'"
*.out plot term 1 vt 1e3 scalex plotVT1Preamble in "vsource.out"

.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Input Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 2 vt 1e3 scalex plotVT1Preamble in "vprimary.out"

.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Output Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 3 vt 1e3 scalex plotVT1Preamble in "vsecondary.out"


.end

```


Simulation Results: Below are results from above sample netlist. With input of 10 Volts the output was stepped-up by a factor of 10, which makes the output voltage equal to 100 Volts.

$n_2 \cdot V_{in} = n_1 \cdot V_{out}$, therefore $V_{out} = V_{in} \cdot (n_2/n_1) = 10V \cdot (10/1) = 100V$

Input Voltage:



Output Voltage:



Sample Netlist: Step down example

```
***** Ideal Transformer Test *****
***** transformer.net
***** transformer In LRT Out LRT n1=turns_primary n2=turns_second

.options gmin=0 ftol=1.e-10
.tran2 tstop = 50e-3 timestep = 10e-6 gcomp=0 im=2
```

```

***** print detailed info to .OUT file
.options verbose

***** Circuit *****
*+++++ source for transient analysis
vsource:v1 1 0 vac=10V f=60
r:rin 1 2 r=50

*+++++for turn ratio 10-to-1 -> Vin/t=Vout: 10V=0.1V
transformer:t1 2 0 3 0 n1=10

r:rload 3 0 r=1e6

***** Simulation *****
.options gnuplot

***** Transient Simulation Results *****
*+++++ set up plot preamble with font and label information
*.options plotVT1Preamble="set term x11 font 'helvetica,18';
*       set title 'Source Voltage';
*       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
*.out plot term 1 vt 1e3 scalex plotVT1Preamble in "vsource.out"

.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Input Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 2 vt 1e3 scalex plotVT1Preamble in "vprimary.out"

.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Output Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 3 vt 1e3 scalex plotVT1Preamble in "vsecondary.out"


.end

```


Simulation Results: Below are results from above step-down sample netlist. With input of 10 Volts output was 1 Volt.

$n_2 \cdot V_{in} = n_1 \cdot V_{out}$, therefore $V_{out} = V_{in} \cdot (n_2/n_1) = 10V \cdot (1/10) = 1V$

Input Voltage:



Output Voltage:



Sample Netlist: Example for using other reference point then ground

```
***** Ideal Transformer Test *****
***** transformer.net
***** transformer In LRT Out LRT n1=turns_primary n2=turns_second

.options gmin=0 ftol=1.e-10
```

```

.tran2 tstop = 50e-3 tstep = 10e-6 gcomp=0 im=2

***** print detailed info to .OUT file
.options verbose

***** Circuit *****
*+++++ source for transient analysis
vsource:v1 1 0 vac=10V f=60
r:r1n 1 2 r=50

*+++++ for output reference other than Ground
*+++++ turn ratio=1-to-1 -> Vin=Vout
transformer:t1 2 0 3 4
*+++++ Define reference point for FREEDA
.ref 4
r:rload 3 4 r=1.e6

***** Simulation *****
.options gnuplot

***** Transient Simulation Results *****
*+++++ set up plot preamble with font and label information
*.options plotVT1Preamble="set term x11 font 'helvetica,18';
*       set title 'Source Voltage';
*       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
*.out plot term 1 vt 1e3 scalex plotVT1Preamble in "vsource.out"

.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Input Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 2 vt 1e3 scalex plotVT1Preamble in "vprimary.out"


.options plotVT1Preamble="set term x11 font 'helvetica,18';
       set title 'Voltage at Output Terminal';
       set xlabel 'Time (milliseconds)'; set ylabel 'Voltage (V_ac)'"
.out plot term 3 vt 1e3 scalex plotVT1Preamble in "vsecondary.out"

.end


```

Simulation Results: Below are results from above sample netlist. With input of 10 Volts output was 10 Volt, because default turn ratio is 1. In the simulator “.ref 4” defines reference point, other than ground. So here the circuit has two independent sides. This netlist also works as step-up and/or step-down voltage transformation for different n1 & n2 values.

Input Voltage:



Output Voltage:



Sample Netlist: Example for using .AC analysis

```
***** Ideal Transformer Test *****
***** transformer.net
***** transformer In LRT Out LRT n1=turns_primary n2=turns_second

.options gmin=0 ftol=1.e-10
```



```

.ac start=1 stop = 100 n_fregs=10
***** print detailed info to .OUT file
.options verbose

***** Circuit *****
*+++++ source for ac analysis
vsource:v1 1 0 vac=10V

r:r1n 1 2 r=50

*+++++for turn ratio 10-to-1 -> Vin/t=Vout: 10V=0.1V
transformer:t1 2 0 3 0 n1=10

r:rload 3 0 r=1e6

***** Simulation *****
.options gnuplot

***** Frequency Simulation Results (.AC analysis)*****
.options plotVT1Preamble="set term x11 font 'helvetica,18';
    set title 'Voltage at Input Terminal';
    set xlabel 'Frequency (Hz)'; set ylabel 'Voltage (V_ac)'"
.out plot term 2 vf mag plotVT1Preamble in "vprimary.out"


.options plotVT1Preamble="set term x11 font 'helvetica,18';
    set title 'Voltage at Output Terminal';
    set xlabel 'Frequency (Hz)'; set ylabel 'Voltage (V_ac)'"
.out plot term 3 vf mag plotVT1Preamble in "vsecondary.out"

.end


```

Simulation Results: Below are results from above sample netlist. With input of 10 Volts the output was 1 Volt, because turn ratio was 10-to-1. And as an ideal transformer output is flat across all frequency.

Input Voltage:



Output Voltage:



Known Bugs:
None found.

Version: 2008.04.15

Credits:

Name	Affiliation	Date	Links
Olga Andreescu oandree@ncsu.edu	NC State University	April 2008	www.ncsu.edu