
Figure 1: Lumped physical model of a wide-band distributed spiral inductor on silicon.  

 
 
Form: lspiral:<instance name> n1 n2 n3 <parameter list> 
 n1, n2, and n3 are the element terminals. 
 Terminal n3 is the element’s reference terminal. 

 
Parameters: 
 

Parameter Type Default value Required? 
rdc: low-frequency series resistance (Ω) DOUBLE n/a yes 
ldc: low-frequency series inductance (H) DOUBLE n/a yes 
rs1: frequency dependent resistance (Ω) DOUBLE n/a yes 
ls1: frequency dependent inductance (H) DOUBLE n/a yes 
ms1:mutual inductance between ldc and 

ls1 (H) DOUBLE n/a yes 

cw:capacitance between metal windings 
of inductor (F) DOUBLE n/a yes 

cox1: oxide capacitance (F) DOUBLE n/a yes 
csub11: substrate capacitance (F) DOUBLE n/a yes 
rsub11: substrate resistivity (Ω) DOUBLE n/a yes 

cox2: oxide capacitance (F) DOUBLE n/a yes 
csub21: substrate capacitance DOUBLE n/a yes 



rsub21: substrate resistivity (Ω) DOUBLE n/a yes 
cox3: oxide capacitance (F) DOUBLE n/a yes 

csub31: substrate capacitance (F) DOUBLE n/a yes 
rsub31: substrate resistivity (Ω) DOUBLE n/a yes 

 
1. coxn,rsubnm and csubnm are process dependent parameters.  
2. The other parameters may obtained analytically or using an appropriate extraction 

procedure [1], [2].  
 

Example: Planar 9.5-nH spiral inductor [1] 
 
lspiral:ls1 1 2 0 rdc=2.778 ldc=4.325e-9 rs1=26e3 ls1=1e-6 ms1=13.61e-9 
+ cw=10.00e-15 cox1=19.71e-15 csub11=64.67e-15 rsub11=101 cox2=73.01e-15 
+ csub21=10.82e-15 rsub21=457 cox3=15.27e-15 csub31=31.06e-15 
+ rsub31=594 

 
Details: 
 
Example of transient analysis for a parallel RLC tank using a 9.5-nH planar spiral 
inductor 
 
netlist file: 
 
* lspiral transient analysis 
.options verbose 
.options gsub11=9.90e-3 gsub21=2.189e-3 gsub31=1.688e-3 
 
.tran2 tstop=1n tstep=4e-13 out_steps=100 
 
isource:isrc 0 1 iac=1m f=8.8GHz 
 
* RLC tank - capacitor value ignores non-idealities 
resistor:RL 1 0 r=1k 
capacitor:CL 1 0 c={1/(9.5e-9*(8.8GHz*2*pi)**2)} 
lspiral:ls1 1 0 0 rdc=2.778 ldc=4.325e-9 rs1=26e3 ls1=1e-6 ms1=13.61e-9 
+ cw=10.00e-15 cox1=19.71e-15 csub11=64.67e-15 rsub11={1/gsub11} cox2=73.01e-15 
+ csub21=10.82e-15 rsub21={1/gsub21} cox3=15.27e-15 csub31=31.06e-15 
+ rsub31={1/gsub31} 
 
.options gnuplot 
 
.options preamble="set title 'Parallel RLC Tank: 9.5-nH Planar spiral inductor'; set zero 1e-15; set 
grid; set data style lines;set xlabel 'Time (s)'; set key noautotitles; set ylabel 'Tank Voltage (V)';set 
yrange [-0.8:0.8]" 
 
.out plot term "1" vt preamble in "lsp_tr.dat" 
 



 
Figure 2: Transient response of a RLC tank modeled with lspiral 

 
 
 
 

Figure 3: Transient response of a RLC tank modeled with an (ideal) inductor 

 
 
Example of ac analysis for extracting inductor’s quality factor from the short circuit 
admittance parameter (Y11):  
 
netlist file:  
 
* lspiral ac analysis 
.options verbose 
.options zo=50 
 
vsource:vin 1 0 vdc=0 vac=1  
vsource:vprobe 2 0 vdc=0 



.options gsub11=9.90e-3 gsub21=2.189e-3 gsub31=1.688e-3 

.ac start=100e6 stop=10GHz n_freqs=10000 
 
lspiral:ls1 1 2 0 rdc=2.778 ldc=4.325e-9 rs1=26e3 ls1=1e-6 ms1=13.61e-9 
+ cw=10.00e-15 cox1=19.71e-15 csub11=64.67e-15 rsub11={1/gsub11} cox2=73.01e-15 
+ csub21=10.82e-15 rsub21={1/gsub21} cox3=15.27e-15 csub31=31.06e-15 
+ rsub31={1/gsub31} 
 
.options gnuplot 
 
.options preamble="set title 'Q11: 9.5-nH Planar spiral inductor'; set zero 1e-15; set grid; set data 
style lines;set yrange [0:20]; set ylabel 'Q11';set xrange [100e6:10e9]; set xlabel 'Frequency (Hz)'; 
set key noautotitles; Q(a,b)=(-b/a)" 
.options postamble="using 1:(Q($2,$3))" 
* y11 - short output, measure i1/v1 
.out plot element "vsource:vin" 0 if -1 mult term "1" vf div postamble preamble in "y11.dat" 

 
Figure 4: AC Analysis: Q11 for 9.5-nH lspiral model 

 
 

Credits: 
Name    Affiliation  Date   Links 
grfeller@ncsu.edu 
 

 
References: 
 
[1] A comprehensive compact-modeling methodology for spiral inductors in silicon-
based RFICs, Watson, A.C.; Melendy, D.; Francis, P.; Kyuwoon Hwang; Weisshaar, A. 
Microwave Theory and Techniques, IEEE Transactions on 
On page(s): 849- 857, Volume: 52, Issue: 3, March 2004 
 
[2] Frequency-independent equivalent-circuit model for on-chip spiral inductors 
Yu Cao; Groves, R.A.; Xuejue Huang; Zamdmer, N.D.; Plouchart, J.-O.; Wachnik, R.A.; 
Tsu-Jae King; Chenming Hu; 
Solid-State Circuits, IEEE Journal of 
Volume 38,  Issue 3,  March 2003 Page(s):419 - 426 


